Crank-slider displacement

Front matter

Copyright

Setup the notebook

Reset the environment

```
Remove["Global`*"];
```

Expressions like **Reduce** do not have an option for assumptions. For that case add kinematic bounding assumptions as a list of equations

```
eqnKinAssumptions = r > 0 && l > r && -π ≤ θ2 < π && -π / 2 ≤ θ3 < π / 2
```

$$r > 0 \;\&\&\; l > r \;\&\&\; -\pi \le \theta_2 < \pi \;\&\&\; -\frac{\pi}{2} \le \theta_3 < \frac{\pi}{2}$$

For **Simplify** and **FullSimplify** a list of assumptions is needed

```
lstKinAssumptions = {r > 0, l > 0, r < l, d > 0, -π ≤ θ2 < π, -π / 2 ≤ θ3 < π / 2}
```

$$\left\{r > 0,\ l > 0,\ r < l,\ d > 0,\ -\pi \le \theta_2 < \pi,\ -\frac{\pi}{2} \le \theta_3 < \frac{\pi}{2}\right\}$$

Worksheet assumptions:

```
$Assumptions = 0 ≤ θ2 < 2 π && 0 ≤ θ3 < 2 π && r ∈ Reals && l ∈ Reals && d ∈ Reals;
```

Without this line, *Mathematica* replaces 1/Cos with Sec and 1/Sin with Csc:

```
$PrePrint = # /. {Csc[z_] :→ 1 / Defer@Sin[z], Sec[z_] :→ 1 / Defer@Cos[z]} &;
```

Create publication-ready equations

```
subPretty = {xp → x_p, θ_2'[t] → ω , dDist → d, mcp → m_cp, mcr → m_cr, θ2 → θ_2}
```

```
{xp → x_p, θ_2'[t] → ω, dDist → d, mcp → m_cp, mcr → m_cr, θ2 → θ_2}
```

Problem setup



Symbol	Definition	Units
$\bar{r}$	Crank Radius (Stroke/2)	mm [in]
$\bar{l}$	Connecting Rod Length (Crank Pin Centerline to Crosshead Pin Centerline)	mm [in]
$\bar{d}$	Distance from Crosshead Pin to Main Bearing Centerline	mm [in]
θ_2	Crank Angle (RPM * time: $\theta_2 = \omega\, t$)	rad
θ_3	Connecting Rod Angle	rad
x_p	Piston Displacement from Top Dead Center (TDC)	mm [in]

$\bar{r}$ Crank radius (stroke/2), mm [in]

$\bar{l}$ Connecting rod length, mm [in]

θ_2 Crank angle, radians (equals $\omega\, t$)

θ_3 Connecting rod angle, radians

$\bar{d}$ Distance from crankshaft centerline to slider pin centerline, mm [in]

x_p Displacement from top-dead center (TDC), mm [in]

ω Rotational speed, radians/second

Setup the vector loop equations

In this system the sum of the vectors must all equal zero:

```
r̄ + l̄ + d̄ == 0
```

```
d̄ + l̄ + r̄ == 0
```

This can be re-written using Euler’s identity, then solved for the distance between crank centerline and crosshead centerline, Abs[$\overline{d}$]:

$$eqn1 = r\, e^{i\,\theta_2} + l\, e^{i\,\theta_3} + d\, e^{i\,\theta_4} == 0$$

$$d\, e^{i\,\theta_4} + e^{i\,\theta_3}\, l + e^{i\,\theta_2}\, r == 0$$

Since θ_4 is fixed at 180 degrees, substitute this value:

$$eqn2 = eqn1 \;/.\; \theta_4 \to \pi$$

$$-d + e^{i\,\theta_3}\, l + e^{i\,\theta_2}\, r == 0$$

$$-d + e^{i\,\theta_3}\, l + e^{i\,\theta_2}\, r == 0\, e^{i}$$

$$-d + e^{i\,\theta_3}\, l + e^{i\,\theta_2}\, r == 0$$

Separate into real and imaginary parts:

```
eqn2a = Simplify[Re[ComplexExpand[eqn2][[1]]],
   Assumptions → {r > 0, l > 0, d > 0, 0 ≤ θ2 < 2 π, 0 ≤ θ3 < 2 π}] == 0
eqn2b = Simplify[Im[ComplexExpand[eqn2][[1]]],
   Assumptions → {r > 0, l > 0, d > 0, 0 ≤ θ2 < 2 π, 0 ≤ θ3 < 2 π}] == 0
```

$$-d + r\,\mathrm{Cos}[\theta_2] + l\,\mathrm{Cos}[\theta_3] == 0$$

$$r\,\mathrm{Sin}[\theta_2] + l\,\mathrm{Sin}[\theta_3] == 0$$

The imaginary part, equation 2b, has only one unknown, the connecting rod angle θ_3, solve for this:

```
eqn3 = Last[Reduce[eqn2b && eqnKinAssumptions, θ3]]
```

$$\theta_3 == -\text{ArcSin}\left[\frac{r\,\text{Sin}[\theta_2]}{l}\right]$$

Equation 3 can be substituted into Equation 2a and solve for the distance from the crank centerline to the crosshead pin centerline, d:

```
eqn4a = eqn2a /. ToRules[eqn3]
```

$$-d + r\,\text{Cos}[\theta_2] + l\sqrt{1 - \frac{r^2\,\text{Sin}[\theta_2]^2}{l^2}} == 0$$

```
eqn4 = Last[Reduce[eqn4a && eqnKinAssumptions, d]]
```

$$d == r\,\text{Cos}[\theta_2] + l\sqrt{\frac{l^2 - r^2\,\text{Sin}[\theta_2]^2}{l^2}}$$

In practice it is more common to reference the displacement to the top dead center (TDC) position so displacement is usually written as:

```
eqnDistTDC = xp == (l + r) - d;
eqnDistTDC = Reduce[eqnDistTDC, d]
```

$$d == l + r - xp$$

Substitute in the replacement expression for **d**:

```
eqn5 = eqn4 /. (eqnDistTDC /. Equal → Rule)
```

$$l + r - xp == r\,\text{Cos}[\theta_2] + l\sqrt{\frac{l^2 - r^2\,\text{Sin}[\theta_2]^2}{l^2}}$$

```
eqnSliderDisp = Last[ Reduce[eqn5 && eqnKinAssumptions, xp]];
eqnSliderDisp = Collect[eqnSliderDisp, r];
eqnSliderDisp /. subPretty
```

$$x_p == l + r\,(1 - \text{Cos}[\theta_2]) - l\sqrt{\frac{l^2 - r^2\,\text{Sin}[\theta_2]^2}{l^2}}$$

Test #1 - Simple crank slider

This test uses round numbers making tracing and debug easier.

```
rIn = Quantity[1, "Inches"];
lIn = Quantity[5, "Inches"];
```

Find the value at **0°**:

```
subTest001 = {r → rIn, l → lIn, θ2 → Quantity[0, "Degrees"]};
eqnSliderDisp /. subTest001
```

$xp == 0\ \text{in}$

Find the value at **90°**:

```
subTest001 = {r → rIn, l → lIn, θ2 → Quantity[90, "Degrees"]};
eqnSliderDisp /. subTest001
```

$xp == \left(6 - 2\sqrt{6}\right)\ \text{in}$

Find the value at **180°**:

```
subTest001 = {r → rIn, l → lIn, θ2 → Quantity[180, "Degrees"]};
eqnSliderDisp /. subTest001
```

$xp == 2\ \text{in}$

Find the value at **270°**:

```
subTest001 = {r → rIn, l → lIn, θ2 → Quantity[270, "Degrees"]};
eqnSliderDisp /. subTest001
```

$xp == \left(6 - 2\sqrt{6}\right)\ \text{in}$